// File: bridge/baby-doge-2/BabyDoge2Bridge.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/security/ReentrancyGuard.sol";

contract BabyDoge2Bridge is Ownable, ReentrancyGuard {
    uint256 public nonce;
    mapping(uint256 => bool) public processedNonces;
    mapping(address => TokenBridge) public tokenBridges;

    struct TokenBridge {
        uint256 minAmount;
        uint256 depositFee;
        uint256 withdrawFee;
        bool isSupported;
        bool isPaused;
        mapping(string => address) receivingTokenAddresses;
        mapping(address => bool) whitelistedAddresses;
    }

    event Deposit(
        address depositingTokenAddress,
        address indexed receivingTokenAddress,
        address indexed sender,
        uint256 amount,
        uint256 indexed nonce,
        string toChain
    );

    event Withdraw(
        address indexed tokenAddress,
        address indexed recipient,
        uint256 amount,
        string fromChain,
        uint256 indexed nonce
    );

    event TokenAdded(address indexed tokenAddress);
    event TokenRemoved(address indexed tokenAddress);

    constructor(
        address initialDepositingTokenAddress,
        string memory initialToChain,
        address initialReceivingTokenAddress
    ) {
        // Add initial token to the supported tokens list
        TokenBridge storage tokenBridge = tokenBridges[
            initialDepositingTokenAddress
        ];
        tokenBridge.isSupported = true;
        tokenBridge.minAmount = 1;
        tokenBridge.depositFee = 0;
        tokenBridge.withdrawFee = 0;
        tokenBridge.receivingTokenAddresses[
            initialToChain
        ] = initialReceivingTokenAddress;
    }

    function deposit(
        address depositingTokenAddress,
        uint256 amount,
        string memory toChain
    ) public nonReentrant {
        TokenBridge storage tokenBridge = tokenBridges[depositingTokenAddress];
        require(tokenBridge.isSupported, "Token not supported");
        require(
            tokenBridge.receivingTokenAddresses[toChain] != address(0),
            "Receiving token address not set for this chain"
        );
        require(
            amount >= tokenBridge.minAmount,
            "Amount less than minimum deposit amount"
        );
        require(
            IERC20(depositingTokenAddress).transferFrom(
                msg.sender,
                address(this),
                amount
            ),
            "Token transfer failed"
        );

        uint256 fee;
        if(!tokenBridge.whitelistedAddresses[msg.sender]) {
            fee = (amount * tokenBridge.depositFee) / 10000;
        }

        uint256 netAmount = amount - fee;
        emit Deposit(
            depositingTokenAddress,
            tokenBridge.receivingTokenAddresses[toChain],
            msg.sender,
            netAmount,
            nonce,
            toChain
        );
        nonce++;
    }

    function withdraw(
        address tokenAddress,
        address recipient,
        uint256 amount,
        string memory fromChain,
        uint256 withdrawNonce
    ) external onlyOwner nonReentrant {
        require(
            !processedNonces[withdrawNonce],
            "Withdrawal already processed"
        );
        processedNonces[withdrawNonce] = true;
        TokenBridge storage tokenBridge = tokenBridges[tokenAddress];
        
        uint256 fee;
        if(!tokenBridge.whitelistedAddresses[recipient]) {
            fee = (amount * tokenBridge.withdrawFee) / 10000;
        }

        uint256 netAmount = amount - fee;
        require(
            IERC20(tokenAddress).transfer(recipient, netAmount),
            "Token transfer failed"
        );
        emit Withdraw(
            tokenAddress,
            recipient,
            netAmount,
            fromChain,
            withdrawNonce
        );
    }

    function addTokenBridge(
        address tokenAddress,
        uint256 minAmount,
        uint256 depositFee,
        uint256 withdrawFee
    ) external onlyOwner {
        TokenBridge storage tokenBridge = tokenBridges[tokenAddress];
        require(!tokenBridge.isSupported, "Token bridge already added");
        tokenBridge.isSupported = true;
        tokenBridge.minAmount = minAmount;
        tokenBridge.depositFee = depositFee;
        tokenBridge.withdrawFee = withdrawFee;
        tokenBridge.isPaused = false;
        emit TokenAdded(tokenAddress);
    }

    function removeTokenBridge(address tokenAddress) external onlyOwner {
        TokenBridge storage tokenBridge = tokenBridges[tokenAddress];
        require(tokenBridge.isSupported, "Token bridge not supported");
        tokenBridge.isSupported = false;
        emit TokenRemoved(tokenAddress);
    }

    function setTokenBridgePauseStatus(address tokenAddress, bool pauseStatus)
        external
        onlyOwner
    {
        TokenBridge storage tokenBridge = tokenBridges[tokenAddress];
        require(tokenBridge.isSupported, "Token bridge not supported");
        tokenBridge.isPaused = pauseStatus;
    }

    function setTokenBridgeMinAmount(address tokenAddress, uint256 minAmount)
        external
        onlyOwner
    {
        TokenBridge storage tokenBridge = tokenBridges[tokenAddress];
        require(tokenBridge.isSupported, "Token bridge not supported");
        tokenBridge.minAmount = minAmount;
    }

    function setTokenBridgeFee(
        address tokenAddress,
        uint256 depositFee,
        uint256 withdrawFee
    ) external onlyOwner {
        TokenBridge storage tokenBridge = tokenBridges[tokenAddress];
        require(tokenBridge.isSupported, "Token bridge not supported");
        tokenBridge.depositFee = depositFee;
        tokenBridge.withdrawFee = withdrawFee;
    }

    function setWhitelistedAddress(
        address tokenAddress,
        address[] memory whitelistedAddresses,
        bool status
    ) external onlyOwner {
        TokenBridge storage tokenBridge = tokenBridges[tokenAddress];
        require(tokenBridge.isSupported, "Token bridge not supported");

        for (uint256 i = 0; i < whitelistedAddresses.length; i++) {
            tokenBridge.whitelistedAddresses[whitelistedAddresses[i]] = status;
        }
    }

    function setReceivingTokenAddress(
        address tokenAddress,
        string memory toChain,
        address receivingTokenAddress
    ) external onlyOwner {
        TokenBridge storage tokenBridge = tokenBridges[tokenAddress];
        require(tokenBridge.isSupported, "Token not supported");
        tokenBridge.receivingTokenAddresses[toChain] = receivingTokenAddress;
    }

    function getTokenBridgeDetails(address tokenAddress)
        external
        view
        returns (
            uint256 minAmount,
            uint256 depositFee,
            uint256 withdrawFee,
            bool isPaused
        )
    {
        TokenBridge storage tokenBridge = tokenBridges[tokenAddress];
        return (
            tokenBridge.minAmount,
            tokenBridge.depositFee,
            tokenBridge.withdrawFee,
            tokenBridge.isPaused
        );
    }

    function getReceivingTokenAddress(address tokenAddress, string memory toChain)
        external
        view
        returns (address)
    {
        TokenBridge storage tokenBridge = tokenBridges[tokenAddress];
        return tokenBridge.receivingTokenAddresses[toChain];
    }

    function isAddressWhitelisted(address tokenAddress, address userAddress)
        external
        view
        returns (bool)
    {
        TokenBridge storage tokenBridge = tokenBridges[tokenAddress];
        return tokenBridge.whitelistedAddresses[userAddress];
    }

    function getNonce() external view returns (uint256) {
        return nonce;
    }

    function isNonceProcessed(uint256 _nonce) external view returns (bool) {
        return processedNonces[_nonce];
    }

    function getTotalTransferFees(
        address depositingTokenAddress,
        uint256 amount,
        string memory toChain,
        address userAddress
    ) public view returns (uint256) {
        TokenBridge storage depositingTokenBridge = tokenBridges[depositingTokenAddress];
        require(depositingTokenBridge.isSupported, "Depositing token not supported");

        address receivingTokenAddress = depositingTokenBridge.receivingTokenAddresses[toChain];
        require(receivingTokenAddress != address(0), "Receiving token address not set for this chain");

        TokenBridge storage receivingTokenBridge = tokenBridges[receivingTokenAddress];

        uint256 totalFees = 0;

        if (!depositingTokenBridge.whitelistedAddresses[userAddress]) {
            uint256 depositFee = (amount * depositingTokenBridge.depositFee) / 10000;
            uint256 withdrawFee = (amount * receivingTokenBridge.withdrawFee) / 10000;
            totalFees = depositFee + withdrawFee;
        }

        return totalFees;
    }

    function getReceivableTokens(
        address depositingTokenAddress,
        uint256 amount,
        string memory toChain,
        address userAddress
    ) public view returns (uint256) {
        uint256 totalFees = getTotalTransferFees(depositingTokenAddress, amount, toChain, userAddress);
        uint256 receivableTokens = amount - totalFees;
        return receivableTokens;
    }

}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be _NOT_ENTERED
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}